

```

10 MODE 1
20 PRINT "Menue":PRINT"====":PRINT:PRINT
30 PRINT"Waehlen Sie : "
40 PRINT:PRINT"Tastaturbelegung      (1)"
45 PRINT:PRINT"Deutscher Zeichensatz (2)"
50 PRINT:PRINT"Y,Z normal belegen    (3)"
60 PRINT:PRINT"Ende                  (4)"
65 a=VAL(INKEY$):IF a<1 OR a>4 THEN 65
70 ON a GOTO 100,600,900,1000
100 KEY 141,"LIST ":KEY DEF 36,1,&6C,&4C,141
110 KEY 142,"RUN"+CHR$(13):KEY DEF 50,1,&72,&52,142
120 KEY 143,"EDIT ":KEY DEF 58,1,&65,&45,143
130 KEY 144,"DATA ":KEY DEF 61,1,&64,&44,144
140 KEY 145,"SAVE "+CHR$(34):KEY DEF 68,1,&73,&53,145
150 KEY 146,"CLS"+CHR$(13):KEY DEF 62,1,&63,&43,146
160 KEY 147,"MODE ":KEY DEF 38,1,&6D,&4D,147
170 KEY 148,"INPUT ":KEY DEF 35,1,&69,&49,148
180 KEY 149,"PAPER ":KEY DEF 64,1,&31,&21,149
190 KEY 150,"PEN " :KEY DEF 65,1,&32,&22,150
200 KEY 151,"INK ":KEY DEF 57,1,&33,&23,151
210 KEY 152,"BORDER ":KEY DEF 56,1,&34,&24,152
220 KEY 153,"LOAD"+CHR$(34):KEY DEF 49,1,&35,&25,153
230 KEY 154,"SPEED WRITE ":KEY DEF 48,1,&36,&26,154
240 KEY 155,"LOCATE ":KEY DEF 41,1,&37,&27,155
250 KEY 156,"CHR$(":KEY DEF 40,1,&38,&28,156
260 KEY 157,"TAB(":KEY DEF 33,1,&39,&29,157
270 KEY 158,"SPC(":KEY DEF 32,1,&30,&5F,158
280 ' *** Anzeige ***
290 MODE 2:WINDOW 1,40,1,22:WINDOW #1,41,80,1,22:WINDOW #2,1,80,23,25
300 PRINT" TASTE"TAB(10)"FUNKTION":PRINT STRING$(18,61)
310 PRINT:PRINT" L"TAB(10)"LIST"
320 PRINT:PRINT" R"TAB(10)"RUN"
330 PRINT:PRINT" D"TAB(10)"DATA"
340 PRINT:PRINT" S"TAB(10)"SAVE"CHR$(34)
350 PRINT:PRINT" I"TAB(10)"INPUT "
360 PRINT:PRINT" C"TAB(10)"CLS"
370 PRINT:PRINT" M"TAB(10)"MODE "
380 PRINT:PRINT" 1"TAB(10)"PAPER "
390 WINDOW SWAP 0,1
400 PRINT:PRINT" 2"TAB(10)"PEN "
410 PRINT:PRINT" 3"TAB(10)"INK "
420 PRINT:PRINT" 4"TAB(10)"BORDER "
430 PRINT:PRINT" 5"TAB(10)"LOAD"CHR$(34)
440 PRINT:PRINT" 6"TAB(10)"SPEED WRITE "
450 PRINT:PRINT" 7"TAB(10)"LOCATE "
460 PRINT:PRINT" 8"TAB(10)"CHR$("
470 PRINT:PRINT" 9"TAB(10)"TAB("
480 PRINT:PRINT" 0"TAB(10)"SPC("
490 WINDOW SWAP 0,2
500 PRINT"Alle Befehle werden mit CTRL + Taste erreicht !":PRINT

```

```

510 PRINT TAB(10)"Weiter mit Leertaste"
520 IF INKEY(47)=-1 THEN 520 ELSE 10
600 ' *** Deutscher Zeichensatz ***
610 RESTORE:SYMBOL AFTER 64:FOR i=1 TO 6:FOR j=1 TO 8
620 READ a(j):NEXT
630 SYMBOL a(1),102,a(2),a(3),a(4),a(5),a(6),a(7),a(8)
640 NEXT
650 DATA 91,0,60,6,62,102,62,0
660 DATA 123,24,60,102,126,102,102,0
670 DATA 93,0,60,102,102,102,102,60,0
680 DATA 125,60,102,102,102,102,60,0
690 DATA 64,0,102,102,102,102,62,0
700 DATA 124,0,102,102,102,102,60,0
710 SYMBOL 163,60,102,102,108,102,102,108
720 ' *** Anzeige ***
730 MODE 1
740 PRINT"Tastenbelegung":PRINT"=====
750 PRINT:PRINT"Tastenummer"TAB(20)"Buchstabe"
760 PRINT STRING$(28,45)
770 PRINT:PRINT"eckige Klammer auf"TAB(22)"["
780 PRINT:PRINT"eckige Klammer zu"TAB(22)"]"
790 PRINT:PRINT"Klammeraffe"TAB(22)"@"
800 PRINT:PRINT" ^"TAB(22)"#"
810 PRINT:PRINT:PRINT"Grossbuchstaben und # mit SHIFT"
820 LOCATE 1,20:PRINT"Vertauschen von Z und Y ? (j/n)"
830 a$=UPPER$(INKEY$)
840 IF a$="N" THEN 10
850 IF a$="J" THEN 860 ELSE 830
860 KEY DEF 71,1,&79,&59 : KEY DEF 43,1,&7A,&5A
870 PRINT:PRINT"Y und Z wurden vertauscht":PRINT:PRINT"Taste druecken"
880 CALL &BB18:GOTO 10
900 KEY DEF 43,1,&79,&59 : KEY DEF 71,1,&7A,&5A
910 CLS
920 LOCATE 1,12:PRINT"Y und Z erreichen Sie wieder ueber die":PRINT:PRINT"normal
en Tasten"
930 LOCATE 10,25:PRINT"Taste druecken":CALL &BB18:GOTO 10
1000 CLS
1010 PRINT:PRINT"Wollen Sie ein neues Programm":PRINT:PRINT"laden (j,n) ?"
1020 a$=UPPER$(INKEY$)
1030 IF a$="J" THEN 1050
1040 IF a$="N" THEN 1100 ELSE 1020
1050 PRINT:CLS:PRINT"Programm wurde geloescht.":PRINT:NEW
1100 CLS:END

```